

To our customers,

---

## Old Company Name in Catalogs and Other Documents

---

On April 1<sup>st</sup>, 2010, NEC Electronics Corporation merged with Renesas Technology Corporation, and Renesas Electronics Corporation took over all the business of both companies. Therefore, although the old company name remains in this document, it is a valid Renesas Electronics document. We appreciate your understanding.

Renesas Electronics website: <http://www.renesas.com>

April 1<sup>st</sup>, 2010  
Renesas Electronics Corporation

Issued by: Renesas Electronics Corporation (<http://www.renesas.com>)

Send any inquiries to <http://www.renesas.com/inquiry>.

# MESC TECHNICAL NEWS No. M16C-15-9807

**M16C/62 Group OTP Version, UV EPROM Version (M30622EC, M30623EC)**

## **Precautions using these products at low voltage**

### **1. Related devices**

M30622EC-XXXFP/GP, M30622ECFS, M30623EC-XXXGP  
(except microcomputers for use in automobiles)

### **2. Symptom**

The initial output level of SOUT3 and SOUT4 are set by the bit 7 of the S I/Oi control register (i= 3, 4) at addresses 0362hex and 0366hex.

However, the following may occur:

- a) When  $V_{cc} < 4.5V$ , the SOUT3 and SOUT4 pins may not be set at the correct logic states.
- b) When SOUTi pin is used as a port pin, the power supply current ( $I_{cc}$ ) may increase as much as 1.4 mA ( $V_{cc}=3.6V$ ). These problems also occur with the M16C/62 emulator pod.

These problems do not occur when  $V_{cc}$  is kept between 4.5V and 5.5V.

### **3. Countermeasures:**

The following sample programs can be used to overcome these problems.

### 3.1 Using the SOUTi pin as a port pin

After the M16C is reset, the following program should be included in the initialization program. See Fig. 1.

- a) Set the SI/Oi bit rate generators at addresses 0363hex and 0367hex to an arbitrary value.
- b) Set the SI/Oi control register to 40hex at addresses 0362hex and 0366hex (do this even if the default value is 40hex after the reset).
- c) Transmit the dummy data (00hex). At this time, SOUTi and CLKi are not output because of the setting in the SI/Oi control registers.
- d) After the dummy transmission is completed, clear the interrupt request bits of SI/Oi (bit 3 of the addresses at 0049hex and 0048hex)
- e) Be sure to disable interrupt during execution of this program.

```

FCLR      I           ; Disable interrupt
MOV.B    #00H, 0363H ; Sets the SI/O3 transfer rate for dummy transmission
BCLR     3, 002CH    ; Disable DMA0 transfer so that no DMA transfer
                    ; occurs during a period from the SI/O3 control
                    ; register protection is released to the time to
                    ; set in the SI/O3 control register.

BCLR     3, 003CH    ; Disable DMA1 transfer
BSET     2, 000AH    ; Release the protection of the SI/O3 control
                    ; register

MOV.B    #40H, 0362H ; Set in the SI/O3 control register
BCLR     6, 035FH    ; Choose SI/O3 as the interrupt factor
BCLR     4, 0049H    ; Set the polarity select bit of the SI/O3 interrupt
                    ; control register to the falling edge.

BCLR     3, 0049H    ; Clear the SI/O3's interrupt request bit
MOV.B    #00H, 0360H ; Dummy transmission

SEND_CHECK:
BTST     3, 0049H    ; Confirm the completion of transmission testing
                    ; the SI/O3's interrupt request bit

JNC      SEND_CHECK ;

BCLR     3, 0049H    ; Clears the SI/O3's interrupt request bit
NOP      ;
NOP      ;
NOP      ;
NOP      ;
NOP      ;
FSET     I           ;

```

Fig.1 Reference program to use the SOUT3 pin as a port

### 3.2 Using SOUTi as the SI/Oi output pin

The following program should be used to initialize the level of SOUTi pin.

#### To initialize SOUTi at the low logic state:

- a) Set the SI/Oi bit rate generators at addresses 0363hex and 0367hex to an arbitrary value.
  - b) Set the SI/Oi control register to 40hex at addresses 0362hex and 0366hex (do this even if the default value is 40hex after the reset).
  - c) Transmit the dummy data (00hex).
  - d) After the dummy transmission is completed, clear the interrupt request bits of SI/Oi. (bit 3 of the addresses at 0049hex and 0048hex)
  - e) Disable interrupt when this program is executed.
- See the sample program in Fig. 2.

#### To initialize SOUTi at the high logic state

- a) The SI/Oi bit rate generators at addresses 0363hex and 0367hex are set at an arbitrary value.
  - f) Set the SI/Oi control register to addresses C0hex at 0362hex and 0366hex (the default value is 40hex after the reset).
  - g) Transmit the dummy data (FFhex).
  - h) After the dummy transmission is completed, clear the interrupt request bits of SI/Oi. (bit 3 of the addresses at 0049hex and 0048hex)
  - i) Disable interrupt when this program is executed.
- See the sample program in Fig. 2.

After the above initialization, then serial I/O communication using SI/O 3 & 4 can start.

```

FCLR      I           ; Disable interrupts
MOV.B     #00H, 0363H ; Set SI/O3 transfer rate
BCLR      3, 002CH    ; Disable DMA0 transfer so that no DMA transfer
                ; occurs during a period from the SI/O3 control
                ; register protection is released to the time to
                ; set in the SI/O3 control register.
BCLR      3, 003CH    ; Disable DMA1 transfer
BSET      2, 000AH    ; Release the protection of the SI/O3 control
                ; register
MOV.B     #40H*, 0362H ; Set in the SI/O3 control register *0C0h is used
                ; for high level.
BCLR      6, 035FH    ; Choose SI/O3 as the interrupt factor
BCLR      4, 0049H    ; Set the bit for switching the SI/O3 interrupt
                ; control register's polarity to the falling edge
BCLR      3, 0049H    ; Clear the SI/O3's interrupt request bit
MOV.B     #00H*, 0360H ; Dummy transmission *0FFh is used for high level.

SEND_CHECK:
BTST      3, 0049H    ; Confirm the completion of transmission owing to
                ; the SI/O3's interrupt request bit
JNC       SEND_CHECK ;
BCLR      3, 0049H    ; Clear the SI/O3's interrupt request bit
;
MOV.B     #10H, 0363H ; Set the SI/O3 transfer rate
BSET      2, 000AH    ; Release the protection of the SI/O3 control
                ; register
MOV.B     #08H, 0362H ; Set in the SI/O3 control register
FSET      I           ;

```

**Fig. 2 A reference program to use SI/O3**

### 3.3 Switching the function of the pin from SOUTi to port

There are two methods to switch the SOUTi to a port:

1. After setting the value of the last serial I/O output bit to the SOUTi initial value (bit 7 at the addresses 0362hex, 0366hex), select the SI/Oi port select bits (bit 3 at the addresses 0362hex, 0366hex).
2. The SOUTi pin initial value is set by the process in "3.2".